Gram-Schmidt, QR Decompostion, and Coordinates.

```
> restart;
  with(LinearAlgebra):
```

Let's run the Gram-Schmidt orthogonalization process on the columns of the following matrix...

```
> A := <<1,4,7>|<2,5,8>|<3,6,9>|<3,9,15>|<1,0,1>>;
```

$$A := \begin{bmatrix} 1 & 2 & 3 & 3 & 1 \\ 4 & 5 & 6 & 9 & 0 \\ 7 & 8 & 9 & 15 & 1 \end{bmatrix} \quad (1)$$

The "GramSchmidt" command takes in a list of vectors. It doesn't like to take a matrix as an input, so I'll pull A apart.

```
> GramSchmidt([seq(A[1..RowDimension(A),i],i=1..ColumnDimension(A))])
  ;
```

$$\left[\begin{bmatrix} 1 \\ 4 \\ 7 \end{bmatrix}, \begin{bmatrix} \frac{9}{11} \\ \frac{3}{11} \\ -\frac{3}{11} \end{bmatrix}, \begin{bmatrix} \frac{1}{3} \\ -\frac{2}{3} \\ \frac{1}{3} \end{bmatrix} \right] \quad (2)$$

This gives unnormalized vectors.

I'll toss in the "normalized" option and convert the list of vectors to a matrix (and call this matrix Q)...

```
> Q := Matrix(GramSchmidt([seq(A[1..RowDimension(A),i],i=1..
  ColumnDimension(A))], normalized));
```

$$Q := \begin{bmatrix} \frac{1}{66}\sqrt{66} & \frac{3}{11}\sqrt{11} & \frac{1}{6}\sqrt{6} \\ \frac{2}{33}\sqrt{66} & \frac{1}{11}\sqrt{11} & -\frac{1}{3}\sqrt{6} \\ \frac{7}{66}\sqrt{66} & -\frac{1}{11}\sqrt{11} & \frac{1}{6}\sqrt{6} \end{bmatrix} \quad (3)$$

Now let R be the matrix whose (i,j)-entry is the dot product of the i-th column of Q dotted with the j-th row of A. In other words, $R = Q^T A$.

```
> R := Transpose(Q).A;
```

$$R := \begin{bmatrix} \sqrt{66} & \frac{13}{11}\sqrt{66} & \frac{15}{11}\sqrt{66} & \frac{24}{11}\sqrt{66} & \frac{4}{33}\sqrt{66} \\ 0 & \frac{3}{11}\sqrt{11} & \frac{6}{11}\sqrt{11} & \frac{3}{11}\sqrt{11} & \frac{2}{11}\sqrt{11} \\ 0 & 0 & 0 & 0 & \frac{1}{3}\sqrt{6} \end{bmatrix} \quad \textbf{(4)}$$

Notice that R is upper-triangular because of the iterative nature of the Gram-Schmidt process. Specifically, the j-th column of R is built from the j-th column of A which can be expressed as a linear combination of the first up to j-th vectors obtained from Gram-Schmidt.

Finally because Q is orthogonal, its inverse is its transpose. Thus $QR = QQ^T A = A$.

```
> Q.R;
```

$$\begin{bmatrix} 1 & 2 & 3 & 3 & 1 \\ 4 & 5 & 6 & 9 & 0 \\ 7 & 8 & 9 & 15 & 1 \end{bmatrix} \quad \textbf{(5)}$$

The following vector belongs to the column space of A:

```
> v := <<2,3,6>>;
```

$$v := \begin{bmatrix} 2 \\ 3 \\ 6 \end{bmatrix} \quad \textbf{(6)}$$

This is revealed by row reduction...

```
> ReducedRowEchelonForm(<A|v>);
```

$$\begin{bmatrix} 1 & 0 & -1 & 1 & 0 & \frac{1}{3} \\ 0 & 1 & 2 & 1 & 0 & \frac{1}{3} \\ 0 & 0 & 0 & 0 & 1 & 1 \end{bmatrix} \quad \textbf{(7)}$$

Notice that v is the 1/3 the first column of A plus 1/3 the second column plus the fifth column of A. This means that relative to the pivot column basis for A, the coordinates of v are...

```
> colCoordsv := <<1/3,1/3,1>>;
```

$$colCoordsv := \begin{bmatrix} \frac{1}{3} \\ \frac{1}{3} \\ 1 \end{bmatrix} \quad (8)$$

Recall that the columns of Q also form a basis for the column space of A. (Gram-Schmidt kicks out a basis - but not just any kind of basis - an orthogonal basis!)

Let's see that v belongs to the column space of A by seeing it expressed in this orthogonal basis....

```
> ReducedRowEchelonForm(<Q|v>);
```

$$\begin{bmatrix} 1 & 0 & 0 & \frac{28}{33}\sqrt{66} \\ 0 & 1 & 0 & \frac{3}{11}\sqrt{11} \\ 0 & 0 & 1 & \frac{1}{3}\sqrt{6} \end{bmatrix} \quad (9)$$

Easier than row reduction, we can find coordinates by using dot products. Each coordinate of v (in this orthonormal basis) is nothing more than the dot product of v with a column of Q.

```
> <<DotProduct(v,Q[1..3,1]),
    DotProduct(v,Q[1..3,2]),
    DotProduct(v,Q[1..3,3])>>;
```

$$\begin{bmatrix} \frac{28}{33}\sqrt{66} \\ \frac{3}{11}\sqrt{11} \\ \frac{1}{3}\sqrt{6} \end{bmatrix} \quad (10)$$